
** Table A3: Gains from Marriage
********************************************************************************
* Regression Analysis
********************************************************************************
use "$OUTDATA/NFHS_sample.dta", clear 


* Macros

* Outcomes

global dec " w_dec n_w_dec sayl hcdec huss saysc "
global viol " injuries n_inj viol_p2 viol_p1 viol_s viol_e "

global marriage "age_marr age_gap educ_gap height_diff workh h_goodjob h_earnless drunk_often"
global wealth "asset1 asset2 modern_cook modern_roof agrland_hec wealth_index sl_index "
global hc "educw primary height lowheight"
global reporting "viol_byothers viol_natal told_anyone sought_help1 sought_help2 sought_help3"

* Covariates and FE
global cov "rural christian hindu scstbc"
global cov1 "wealth_index hhsize educw"
global cov2 "ageh educh workw workh"
global fe1   "i.state i.yy_1marr"
global fe2   "i.state i.yearb"
global did "i.dow_cohort##i.hindu1"

global cluster "state"

global graph "ylabel(, nogrid) scheme(s2mono) plotregion(fcolor(white) lcolor(white) margin(small)) graphregion(fcolor(white) lcolor(white) ifcolor(white) ilcolor(white))"


* Labels for Tables

label var w_dec "\shortstack{Any \\ Decision}"
label var n_w_dec "\shortstack{No. of \\ Decisions}"
label var sayl "\shortstack{Household \\ Purchases}"
label var hcdec "\shortstack{Health \& \\ Contracept.}"
label var huss "\shortstack{Husband's \\ Money}"
label var saysc "\shortstack{Daily \\ Decisions}"

label var injuries "\shortstack{Any \\ Injury}"
label var n_inj "\shortstack{No. of \\ Injuries}"
label var viol_p2 "\shortstack{Severe \\ Violence}"
label var viol_p1 "\shortstack{Less Severe \\ Violence}"
label var viol_s "\shortstack{Sexual \\ Violence}"
label var viol_e "\shortstack{Emotional \\ Violence}"

label var age_marr "\shortstack{Age at \\ Marriage}"
label var age_gap "\shortstack{Spousal \\ Age Gap}"
label var educ_gap "\shortstack{Spousal \\ Educ. Gap}" 
label var height_diff "\shortstack{Absolute \\ Height Gap}" 
label var workh "\shortstack{Husband \\ Employed}" 
label var h_goodjob "\shortstack{Husband \\ White Collar \\ Job}"   
label var h_earnless "\shortstack{Husband \\Earns Less \\ or Same}" 
label var drunk_often "\shortstack{Husband \\Often Drunk}"

label var educw "\shortstack{Years of \\ Schooling}"
label var primary "\shortstack{Primary \\ School}"
label var height "\shortstack{Height \\(cm)}"
label var lowheight "\shortstack{Low \\ Height}"

label var viol_byothers "\shortstack{Viol. \\ by Others}"
label var viol_natal "\shortstack{Viol. in \\ Natal Family}"
label var told_anyone "\shortstack{Has Told \\ Anyone }"
label var sought_help1 "\shortstack{Help \\ From Family}"
label var sought_help2 "\shortstack{Help From \\ Friends \& \\ Network}"
label var sought_help3 "\shortstack{Help From \\ Police\/Doctor}"

label var scstbc "SC/ST/OBC"
label var rural "Rural"
label var muslim "Muslim"
label var christian "Christian"
label var hindu "Hindu"
label var wealth_index "Wealth Index"
label var hhsize "Household Size"
label var workw "Employed in Past 12 Months"
label var nkids "No. Kids"
label var nkids_home "No. Kids at Home"
label var nomorec "Completed Fertility"
label var kids_home "Kids at Home"


eststo clear

* Heterogeneity in gains from marriage

label var w_dec "\shortstack{Any \\ Decision}"
label var injuries "\shortstack{Any \\ Injury}"
label var nkids_home "No. Kids"
label var kids_home "Any Kid"

cap gen firstbornboy = (b4_01 == 1)
replace firstbornboy = . if b4_01 == .
label var firstbornboy "First Born Boy"

foreach y in w_dec injuries {

	eststo: reg `y' i.dowTreat##c.nkids_home dow_cohort hindu1 $cov $fe2 [pweight = v005]  , r cl($cluster )
	qui sum `y' if e(sample)
	estadd scalar m =r(mean)
	
	eststo: reg `y' i.dowTreat##i.nomorec dow_cohort hindu1 $cov $fe2 [pweight = v005]  , r cl($cluster )
	qui sum `y' if e(sample)
	estadd scalar m =r(mean)
	
	eststo: reg `y' i.dowTreat##i.firstbornboy dow_cohort hindu1 $cov $fe2 [pweight = v005]  , r cl($cluster )
	qui sum `y' if e(sample)
	estadd scalar m =r(mean)
	
	
	}
	
esttab using "$OUTDIR/NFHS3_maritalsurplus.tex", ///
booktabs nonotes replace indicate("\hline\vspace{-3mm} \\ \vspace{-3mm}  Individual Controls = $cov " " \vspace{-3mm} State FE = *.state " " Year of Birth FE = *.yearb ", labels("Yes" "$-$") ) ///
keep(1.dowTreat 1.dowTreat#c.nkids_home  1.dowTreat#1.nomorec 1.dowTreat#1.firstbornboy) ///
label interaction(" \$\times\$ ") substitute("=1" "") ///
stats(N r2 m, labels("Obs." "R sq." "Mean Dep. Var.") fmt(%9.0fc %9.3fc))   ///
se star(* 0.10 ** 0.05 *** 0.01)  b(%9.3fc) se(%9.3fc) 

est clear
eststo clear
global keyoutcomes "sayl viol_p2"
cls 
foreach y in $keyoutcomes {
	
	eststo `y'_nkids: reg `y' i.dowTreat##c.nkids_home dow_cohort hindu1 $cov $fe2 [pweight = v005]  , r cl($cluster )
	quietly estadd local yb "Yes", replace
	quietly estadd local state "Yes", replace 
	quietly estadd local cov "Yes", replace 
	qui sum `y' if e(sample)
	estadd scalar m =r(mean)
	lincom  1.dowTreat#c.nkids_home + 1.dowTreat
		estadd scalar bval`y'_nkids = r(estimate)
		estadd scalar bpval`y'_nkids = r(p)
}		
	    
foreach y in $keyoutcomes {	    
	eststo `y'_nomorec: reg `y' i.dowTreat##i.nomorec dow_cohort hindu1 $cov $fe2 [pweight = v005]  , r cl($cluster )
	quietly estadd local yb "Yes", replace
	quietly estadd local state "Yes", replace 
	quietly estadd local cov "Yes", replace 
	qui sum `y' if e(sample)
	estadd scalar m =r(mean)
	lincom  1.dowTreat#1.nomorec + 1.dowTreat
		estadd scalar bval`y'_nomorec = r(estimate)
		estadd scalar bpval`y'_nomorec = r(p)
		
}

foreach y in $keyoutcomes {
	
	eststo `y'_firstbornboy: reg `y' i.dowTreat##i.firstbornboy dow_cohort hindu1 $cov $fe2 [pweight = v005]  , r cl($cluster )
	quietly estadd local yb "Yes", replace
	quietly estadd local state "Yes", replace 
	quietly estadd local cov "Yes", replace 
	qui sum `y' if e(sample)
	estadd scalar m =r(mean)
	lincom  1.dowTreat#1.firstbornboy + 1.dowTreat
		estadd scalar bval`y'_firstbornboy = r(estimate)
		estadd scalar bpval`y'_firstbornboy = r(p)
		
}

esttab sayl_nkids sayl_nomorec sayl_firstbornboy viol_p2_nkids viol_p2_nomorec viol_p2_firstbornboy using "$OUTTEX/keyoutcomes_gains.tex", replace label se star(* 0.10 ** 0.05 *** 0.01) s(bvalsayl_nkids bpvalsayl_nkids bvalviol_p2_nkids bpvalviol_p2_nkids bvalsayl_nomorec bpvalsayl_nomorec bvalviol_p2_nomorec bpvalviol_p2_nomorec bvalsayl_firstbornboy bpvalsayl_firstbornboy bvalviol_p2_firstbornboy bpvalviol_p2_firstbornboy yb state cov N m ,label("Treated + Treated x Number of Chidren(Coefficient)" "Treated + Treated x Number of Chidren(p-value)" "Treated + Treated x Number of Chidren(Coefficient)" "Treated + Treated x Number of Chidren(p-value)" "Treated + Treated x First Born Boy(Coefficient)" "Treated + Treated x First Born Boy(p-value)" "Treated + Treated x Number of Chidren(Coefficient)" "Treated + Treated x Number of Chidren(p-value)" "Treated + Treated x Number of Chidren(Coefficient)" "Treated + Treated x Number of Chidren(p-value)" "Treated + Treated x First Born Boy(Coefficient)" "Treated + Treated x First Born Boy(p-value)" "Year of Birth Fixed Effects" "State of Residence Fixed Effects" "Covariates" "Observations" "Mean of Dep. Variable") fmt(%9.3fc %9.3fc %9.3fc %9.3fc %9.3fc %9.3fc %9.3fc %9.3fc %9.3fc %9.3fc %9.3fc %9.3fc %9.0fc %9.0fc %9.0fc %9.0fc %9.3fc) ) keep(1.dowTreat 1.dowTreat#c.nkids_home 1.dowTreat#1.nomorec 1.dowTreat#1.firstbornboy) b(%9.3fc) se(%9.3fc) 
